Excel® 2019 Functions & Formulas

Glossary



1. Formula: An equation in a cell that results in a new value.
2. Function: Pre-built formula that performs an operation on values resulting in one or more values.
3. Argument: The values used by a function. Multiple arguments separated by commas. Can be another function (creating a nested function).
4. Constant: An unchanging value.
5. Operator: Symbol specifying a calculation to perform.
   See **Controlling Order of Precedence** for more information.
6. Reference: A cell, range of cells, or a name that represents a value or cell(s). The order of precedence for references is:
   : (colon) Range.
   _ (space) Intersection.
   , (comma) Union.
7. Boolean or logical value, TRUE or FALSE.

Controlling Order of Precedence

The order of precedence for symbols is:
Arithmetic: - Negative, % Percent, ^ Exponentiation, * and / Multiplication and Division, + and - Addition and Subtraction.
Comparison: = Equals, < Less than, > Greater than, <= Less than or equal to, >= Greater than or equal to, <> Does not equal. Use parentheses to change order.
E.g., = (TODAY() – InvoiceDate) / 30

Conditionally Summing/Counting Data

The traditional SUM() and COUNT() functions can be extended with SUMIF() and COUNTIF() to only add or enumerate cells that match certain criteria.

=SUMIF(range, criteria, sum_range)

Sums only numbers in ***range*** that match ***criteria***. Optional ***sum_range*** allows cells summed to be in a different range than cells matching criteria.
E.g., = SUMIF(B1:B5,"<0") results in -8 if cells B1 through B5 contain 1, -2, 3,-6, 10.

=COUNTIF(range, criteria)

Counts text entries or numbers in ***range*** that match ***criteria***.

Being Precise (Rounding)

Use Excel's various rounding functions when cells display slightly different values than what is actually stored. For example, Excel may display 5.12 due to formatting for two decimal places, but the number actually stored in the cell and used in calculations could be 5.1242823.

Rounding Up or Down to a Multiple

=ROUND(number, number of digits)

Rounds to nearest integer, decimal place or multiple.
Rounds a ***number*** to specified ***number of digits*** where ***number*** can refer to a cell and ***number of digits*** can be:
0 - round to nearest integer.
Positive - round to required decimal places.
Negative - round to required multiples.
E.g., =ROUND(3.4376, 2) rounds 3.4376 to 2 decimal places or 3.44.
=ROUND(67562.34, -2) rounds 67,562.34 to the nearest 100, or 67,600.

=MROUND(number, multiple)

Rounds to a specified multiple.
Rounds a positive ***number*** up and away from zero if the remainder of dividing ***number*** by ***multiple*** is >= half of ***multiple***. Also rounds up and away from zero if negative, producing a larger negative.
A positive ***number*** must have a positive ***multiple*** and a negative ***number*** a negative ***multiple***.
E.g., =MROUND (3.4376, 2) rounds 3.4376 up to the nearest multiple of 2 returning 4.
=MROUND (-67562.34, -0.2) rounds -67,562.34 up to the nearest multiple of -0.2 returning -67,562.4.

Rounding Only One Way

=ROUNDUP(number, number of digits)

Same as ROUND but always rounds up and away from 0. ROUNDDOWN always rounds down and toward 0.

=CEILING(number, significance)

Same as MROUND except always rounds up unless ***number*** is negative and ***significance*** is positive. E.g.,
=CEILING(10.1, 2) = 12
=CEILING(-9.9, 2) = -8
=CEILING(-10.1, -2) = -12
FLOOR is a similar function that rounds down.

=CEILING.PRECISE(number, significance)

ISO version which always rounds up. FLOOR.PRECISE is the same as FLOOR but always rounds down.

=EVEN(number)

Same as ROUND but always rounds away from zero to the nearest even integer. ODD always rounds away from zero to the nearest odd integer.

=INT(number)

Rounds ***number*** down to the nearest integer.

Improving Clarity with Range Names

Use range names to help understand and manage cell references. Functions are more understandable if references are descriptive. For example, the intention of =AVERAGE(PurchasePrice) is clearer than =AVERAGE(B2:B7).

Name Rules

- Start with a letter, underscore or backslash.
- Remaining characters may be alphanumeric, a period or an underscore.
- Do not use: spaces, cell references (e.g., A100, R2C3), or just "C", "c", "R", or "r".
- Maximum 255 characters.
- Names are not case sensitive.

Creating a Name

1. Select the cell(s) to name.
2. Choose FORMULAS, then Define Name, or [Click] in the Name Box.



The Name Box shows the active cell address or, if defined, the range name.

3. Type a name for the range and press <Enter>.

Limiting the Scope of a Name

Use scope to limit the worksheet(s) within which the name is valid allowing the same name to be used on different worksheets. To limit scope to a single worksheet:

1. Choose FORMULAS, then Define Name.
2. Fill in details. Specify SCOPE, then [Click] OK.

The cells referred to can be in a different worksheet than the scope.

Defining a Constant or Formula for a Name

1. Choose FORMULAS, then Define Name.
2. Type the NAME, SCOPE, and optionally COMMENT.
3. In REFERS TO, type the constant or formula (preceded by =). [Click] OK.

Creating Several Names at Once

Row or column labels can be used to create named ranges. Labels can be above, below, left or right of the data.

1. Select existing data and the labels.
2. Choose FORMULAS, then Create from Selection.
3. Specify which part of the selection to use as names, e.g., TOP ROW and/or LEFT COLUMN. [Click] OK.

For example:

	A	B	C	D
1	YTD Sales	Qtr1	Qtr2	Qtr3
2	East	122	351	844
3	Central	213	531	399
4	West	866	429	557

Creates 7 named ranges. One for each column and row of data, plus a range called YTDSales for B2 through D4.

Examples of Summing Selected Data

A toy manufacturer, Happy Toys, uses SUMIF to track worldwide shipment costs by country.

	B	C	D
2	Country	Shipping Cost	TOTALS
3	US	1,098.00	US shipping
4	Canada	3,674.00	$3,571.00
5	US	2,473.00	
6	India	8,674.00	
7	Japan	10,632.00	
8	India	8,980.00	

CELL D4 is $3,571.00 because =SUMIF(B3:B8, "US", C3:C8) finds all "US" cells in B3:B8 then sums their C3:C8 counterparts.

Examples of Being Precise

Happy Toys wants to set Canadian prices. Multiplying US prices by the exchange rate may not be market smart. Rounding up is always better for the bottom line so ROUND and MROUND are abandoned in favor of CEILING. Additional tinkering is required to satisfy Marketing's desire for Canadian prices to always end in 99.

CELL B6: =B3*C3 results in 19.550772 which is formatted to 19.551 with 3 decimals.
CELL B7: =ROUND(B6,1) results in 19.6 which is formatted to 19.60 with 2 decimals.
CELL B8: =MROUND(B6,0.5) results in 19.50. Cell B6 is rounded to the nearest 50 cents.
CELL B9: =CEILING(B6,0.5) results in 20.00. Cell B6 is rounded **up** to the nearest 50 cents.
CELL B10: =CEILING(B6,0.5)-0.01 results in 19.99. One penny is subtracted from the result. This is the price selected by Marketing.

	B	C	D
2	US Price:	Ex Rate:	
3	19.19	1.0188	
4			
5	Canadian Prices	Function Used	
6	19.551		
7	19.60	ROUND	
8	19.50	MROUND	
9	20.00	CEILING	
10	19.99	CEILING	